

CHUYÊN ĐỀ 3: XỬ LÝ CÔNG THỨC RỄ NHÁNH PHỨC TẠP. KỸ THUẬT LỒNG GHÉP HÀM TRONG CÔNG THỨC, ỨNG DỤNG CÁC HÀM EXCEL QUAN TRỌNG

CÔNG THỨC RỄ NHÁNH


Cấu trúc hàm IF

IF(logic, Value1, Value2)

Hàm IF trả về Value1 nếu logic = True, ngược lại là Value 2. Value1, Value2 có thể hằng số hoặc một biểu thức. Nếu có nhiều điều kiện (nhiều “nếu”) thì Value1, Value thường là một hàm IF để xét tiếp các trường hợp khác nữa.

Cách dùng 1:

=IF(lg1,value1,IF(lg2,value2,IF(lg3,value3, value4)))




Ví dụ 1: Nếu chức vụ (từ ô D4) là “GD” thưởng 200; “PGD” thưởng 150; “TP” thưởng 100, còn lại thưởng 50. Đây là dạng bài toán mà ứng với mỗi điều kiện trả về một giá trị tương ứng (nhiều điều kiện, nhiều giá trị).

=IF(D4=”GD”,200,IF(D4=”PGD”,150,IF(D4=”TP”,100,50)))

Cách dùng 2:

=IF(lg1,IF(lg2,IF(lg3,value3, 0),0),0) Nếu công thức để lấy văn bản thì 0 đổi thành ” ”




Ví dụ 2: Nếu chức vụ (từ ô D4) là “GD” và ngày công >22 (ô E4) thưởng 200. Đây là dạng bài xét nhiều điều kiện rồi mới trả về một giá trị cuối cùng (nhiều điều kiện, một giá trị).

=IF(D4=”GD”,IF(E4>22, 200,0),0)

Thực ra công thức này giống cách dùng hàm AND nhưng trong kỹ thuật nâng cao thì không dùng hàm AND, hoặc dùng nó lại là sai.

Cách dùng 3: Tổng các IF nếu kết quả trả về có thể một hoặc nhiều kết quả



Ví dụ 3: Nếu chức vụ (từ ô D4) là “GD” thưởng 200; nếu ngày công >22 (ô E4) thưởng 100; nếu sản xuất = “HT” (ô F4) thưởng 80. Đây là dạng bài xét công thức trả về từ 1-n giá trị. Tức là không phải chỉ lấy 1 giá trị như hai cách trên. Ta dùng phương pháp tổng logic.

=IF(D4=”GD”, 200,0) + IF(E4>22, 100,0) + IF(F4=”HT”, 80,0)

(Với yêu cầu bài này mà dùng phương pháp IF lồng IF là sai).

KỸ THUẬT LỒNG GHÉP CÔNG THỨC, ỨNG DỤNG CÁC HÀM QUAN TRỌNG TRONG EXCEL

Phương pháp lập công thức rẽ nhánh theo khoảng giá trị

- + Cách 1: dùng AND(logic cận dưới, logic cận trên)
- + Cách 2: so sánh < cận trên, so sánh từ nhỏ->lớn - pp loại trừ cận dưới
- + Cách 3: so sánh > cận dưới, so sánh từ lớn->nhỏ - pp loại trừ cận trên

Các hàm ứng dụng:

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

Hàm tìm lookup_value tại cột đầu tiên trong bảng/mảng table_array, nếu tìm được sẽ trả về giá trị ở dòng tương ứng tại cột ở vị trí col_index_num trong bảng.

+ range_lookup: nếu là TRUE/1 (ngầm định) hàm tìm giá trị lớn nhất mà <= lookup_value, cột đầu tiên của table_array phải được sắp xếp tăng dần. Nếu là FALSE tìm tuyệt đối khi đó table_array không cần sắp xếp.

	A	B	C	D	E	F	G
10	Ten	Diem	Xep loai		Điều kiện	ASC	Đánh giá
11	a	1	Kem		0-<5	0	Kem
12	b	2	Kem		5-<7	5	TB
13	c	5	TB		7-<8	7	Kha
14	d	7	Kha		>=8	8	Gioi
15	g	2	=VLOOKUP(B15,\$F\$11:\$G\$14,2,TRUE)				
16	h	10	Gioi				
17	f	8	Gioi				

HLOOKUP(lookup_value, table_array, row_index_num, [range_lookup])

Hàm dùng tương tự như VLOOKUP nhưng tìm lookup_value trên dòng đầu tiên. Nếu tìm được thì trả về giá trị ở dòng row_index_num tại cột tương ứng tìm được.

LOOKUP(lookup_value, lookup_vector, [result_vector])

+ lookup_vector: vùng tìm kiếm cho lookup_value, vùng chỉ được một dòng hoặc một cột, phải được sắp xếp tăng dần.

+ result_vector: vùng kết quả, giá trị trong vùng này được trả về ở vị trí tìm được tương ứng của lookup_vector. lookup_vector và result_vector phải cùng kích thước, ô đầu tiên của vùng có thể khác nhau, hai vùng không nhất thiết phải trong cùng một bảng.

	A	B	C	D	E	F	G
18					1		
19					2		
20					3		
21					4		V1
22					5		V2
23					6		V3
24		LOOKUP	=LOOKUP(3.5,E18:E23,G21:G26)				V4
25							V5
26							V6

Cấu trúc 2 của hàm LOOKUP

LOOKUP(lookup_value, array)

Hàm tìm lookup_value tại cột đầu tiên trong mảng array, nếu tìm được trả về giá trị ở cột cuối cùng của mảng array.

	A	B	C	D	E	F	G
9			VLOOKUP	LOOKUP			
10	Ten	Diem	Xep loai	Xep loai	Điều kiện	ASC	Đánh giá
11	a	1	Kem	Kem	0-<5		0 Kem
12	b	2	Kem	Kem	5-<7		5 TB
13	c	5	TB	TB	7-<8		7 Kha
14	d	7	Kha	Kha	>=8		8 Gioi
15	g	2	Kem	=LOOKUP(B15,\$F\$11:\$G\$14)			
16	h	10	Gioi	Gioi			
17	f	8	Gioi	Gioi			

MATCH(lookup_value, lookup_array, [match_type])

Hàm tìm lookup_value trong mảng/vùng lookup_array, nếu tìm được trả về vị trí của giá trị tìm được trong mảng.

match_type: 1 – ngầm định, tìm tương đối theo lookup_array được sắp xếp tăng dần (tìm giá trị lớn nhất mà nó $\leq$ lookup_value) ; -1 là tìm tương đối theo lookup_array được sắp xếp giảm dần; 0 tìm tuyệt đối.

SUM x ✓ fx =MATCH(3.5,E18:E23,1)					
	A	B	C	D	E
18					1
19					2
20					3
21					4
22					5
23					6
24		LOOKUP	V3		
25		MATCH	=MATCH(3.5,E18:E23,1)		

INDEX(array, row_num, [column_num])

Hàm Index lấy một hoặc mảng giá trị tại dòng- row_num, cột- column_num trong mảng array.

E2 fx =INDEX(A1:C4,2,3)					
	A	B	C	D	E
1	1	2	3		
2	2	3	4		4
3	3	4	5		
4	4	5	6		

=INDEX(A1:C4,2,3) kết quả là 4

Muốn lấy toàn bộ mảng giá trị tại dòng 3:

(muốn hiển thị đầy đủ kết quả của mảng, chọn 3 ô liên tiếp theo dòng)

=INDEX(A1:C4,3,0) nhấn CTRL+SHIFT+ENTER kết quả là {3,4,5}

Muốn lấy toàn bộ mảng giá trị tại cột 2:

(muốn hiển thị đầy đủ kết quả của mảng, chọn 4 ô liên tiếp theo cột)

=INDEX(A1:C4,0,2) nhấn CTRL+SHIFT+ENTER kết quả là {2,3,4,5}

Cấu trúc 2 của INDEX

INDEX(reference, row_num, [column_num], [area_num])

Các dùng gần giống như cách 1 của hàm.

+ Reference: có thể là một vùng hoặc nhiều vùng (area), nếu nhiều vùng thì các vùng đó ngăn cách nhau bởi dấu phẩy (,) và tất cả các vùng được bao trong ngoặc đơn.

+ area_num: để chỉ ra vùng thứ mấy được dùng để lấy ra giá trị hoặc mảng ở tọa độ row_num, column_num.

	A	B	C	D	E	F	G
1	1	2	3			2	
2	2	3	4		4	3	
3	3	4	5			4	
4	4	5	6			5	
5					=INDEX((A1:C4,E2,F1:F4),2,1,3)		

Theo ví dụ trên Reference gồm 3 vùng: A1:C4, C2, F1:F4. Hàm sẽ lấy giá trị ở dòng 2, cột 1 của vùng thứ 3 là F1:F4, kết quả là 3.

=INDEX((A1:C4,E2,F1:F4),0,2,1) nhấn CTRL+SHIFT+ENTER kết quả là {2,3,4,5}

OFFSET(reference, rows, cols, [height], [width]) :

Hàm Offset trả về tham chiếu được điều chỉnh từ tham số reference.

+ reference: là tham chiếu cơ sở để tính toán và điều chỉnh rồi được trả về bởi hàm Offset. Tham số này có thể là địa chỉ một ô hoặc một vùng.

Rows: tăng, giảm dòng đầu tiên của tham chiếu reference

Cols: tăng, giảm cột đầu tiên của tham chiếu reference

Height: số dòng cần lấy trong reference, nếu để trống thì lấy số dòng của reference

width: số cột cần lấy trong reference, nếu để trống thì lấy số cột của reference.

Ví dụ:

	A	B	C	D	E	F	G
1	1	2	3				
2	2	3	4				
3	3	4	5				
4	4	5	6				
5							
6	=SUM(OFFSET(A1:C4,1,1,2,2))						
7							

INDIRECT(ref_text, [a1])

Hàm Indirect trả về tham chiếu từ văn bản tham chiếu.

+ ref_text: là giá trị văn bản mô tả tên/Name hoặc địa chỉ/Address của tham chiếu.

+ a1: nếu là TRUE thì hàm kiểm tra cấu trúc địa chỉ dạng tên cột, số dòng. Nếu FALSE thì hàm kiểm tra cấu trúc địa chỉ dạng C1R1.

Ví dụ:

SUM											=SUM(INDIRECT(B6))
	A	B	C	D	E	F					
1	1	2	3								
2	2	3	4								
3	3	4	5								
4	4	5	6								
5											
6	16	B2:C3									
7	=SUM(INDIRECT(B6))										

BÀI TẬP

1. Thực hành bài toán làm lương
2. Thiết kế hệ thống điểm danh

CHUYÊN ĐỀ 4: KỸ THUẬT LẬP CÔNG THỨC MẢNG - PHÉP TÍNH CÓ NHIỀU ĐIỀU KIỆN PHỨC TẠP

Trong lập báo cáo tính các chỉ tiêu tổng hợp kết hợp nhiều điều kiện. Bình thường chúng ta sử dụng các hàm SumIf, CountIf chỉ với một điều kiện đơn lẻ, nhưng trong thực tế phải kết hợp rất nhiều điều kiện với nhau nên kỹ thuật lập công thức mảng hoặc ứng dụng hàm SumProduct giúp ta làm tốt việc này.

Hàm SUMPRODUCT:

Cấu trúc:

SUMPRODUCT(array1, [array2], [array3], ...)

Array - Mảng dữ liệu là một tập hợp dãy giá trị liên tiếp trong một hàng nào đó. VD A1:C1 hoặc A1:A10,...

Phép tính này cho phép chúng ta tính tổng của tích array1*array2*array3*...*array255.

Ví dụ có số liệu sau:

	A	B	C	D
1	2	20	Cam	Giống lại
2	3	10	Bưởi	Không
3	4	25	Cam	Không

Bây giờ cần tính doanh thu của các loại hoa quả

array1=A1:A3

array2=B1:B3

Công thức =SumProduct(A1:A3, B1:B3) = 170

Bản chất công thức làm việc như sau

=A1*B1+A2*B2+A3*B3 kết quả là 170

Nhắc lại về phép tính logic:

Giá trị kiểu logic chỉ cho ra 1 trong 2 giá trị là TRUE/1 (<>0), FALSE/0

Phép toán logic:<, >, <>, =, >=, <=, Not()

Ví dụ:

2>3=False

3>1=True

4>3=True

(*) Logic VÀ - AND

$= (2 > 3) * (3 > 1) * (4 > 3) = \text{False} * \text{True} * \text{True} = 0 * 1 * 1 = \text{False} / 0$ tương đương với hàm $\text{AND}(2 > 3, 3 > 1, 4 > 3)$. Ít nhất một logic = False thì kết quả sẽ là False hay 0.

* Logic hoặc - OR

$= (2 > 3) + (3 > 1) + (4 > 3) = \text{False} + \text{True} + \text{True} = 0 + 1 = \text{True} / 1$ (hay một số $\neq 0$) tương đương với hàm $\text{OR}(2 > 3, 3 > 1, 4 > 3)$. Ít nhất một logic = True thì kết quả sẽ là True/1 (hay một số $\neq 0$).
Lưu ý tổng của các giá trị là $\text{True} = \text{True} =$ một số $\neq 0$.

*) Tính tổng có nhiều điều kiện:

Dùng SUMPRODUCT

Tính tổng doanh thu của loại là "Cam"

$= \text{SUMPRODUCT}(A1:A3, B1:B3 * (C1:C3 = \text{"Cam"}))$ hoặc

$= \text{SUMPRODUCT}(A1:A3 * B1:B3 * (C1:C3 = \text{"Cam"})) = 140$

Công thức tính như sau:

$= A1 * B1 * (C1 = \text{"Cam"}) + A2 * B2 * (C2 = \text{"Cam"}) + A3 * B3 * (C3 = \text{"Cam"})$

$= 2 * 20 * \text{True} + 3 * 10 * \text{False} + 4 * 25 * \text{True}$

$= 2 * 20 * 1 + 3 * 10 * 0 + 4 * 25 * 1 = 140$

MẢNG VÀ CÔNG THỨC MẢNG – FORMULA ARRAY

Mảng là gì? Mảng là tập hợp các phần tử nằm cạnh nhau liên tiếp. Mảng có thể một chiều hoặc hai chiều (gồm dòng và cột). Trong Excel chia ra là mảng phân bố theo chiều ngang (các cột) hoặc chiều dọc (các dòng).

Là công mảng để nhận mảng cần nhấn **CTRL+SHIFT+ENTER**

Mảng một chiều:

+ Mảng ngang: Tập hợp các phần tử phân bố theo cột. Công thức mô tả như sau

$= \{1, 2, 3\}$

B3		fx {=1,2,3}			
	A	B	C	D	E
2	Mảng 1 chiều				
3	Ngang	1	2	3	

+ Mảng dọc: Tập hợp các phần tử phân bố theo dòng. Công thức mô tả như sau

B4		fx {={"a";"b";"c"}}			
	A	B	C	D	E
2	Mảng 1 chiều				
4	Dọc	a			
5		b			
6		c			

Mảng hai chiều – 2D: là tập hợp các phần tử phân bố cả dòng và cột

B8		fx {={1,"a";2,"b";3,"c"}}				
	A	B	C	D	E	F
7	Mảng 2 chiều					
8		1	a			
9		2	b			
10		3	c			

Lấy cang từ vùng địa chỉ ô

	A	B	C	D	E
12	Dữ liệu gốc				
13		1	2	3	4
14		2	3	4	5
15		3	4	5	6
16					
17	Lấy cả mảng				
18		1	2	3	4
19		2	3	4	5
20		3	4	5	6

CÔNG THỨC MẢNG- PHÉP TÍNH CÓ NHIỀU ĐIỀU KIỆN

	A	B	C	D
1	2	20	Cam	Giống lại
2	3	10	Bưởi	Không
3	4	25	Cam	Không

=Sum(IF(C1:C3="Cam",A1:A3*B1:B3,0))

Kết thúc nhấn tổ hợp phím **CTRL+SHIFT+ENTER**. Với cách làm này EXCEL sẽ phân tích như sau:

Xét trên từng dòng trong mảng (array)

dòng1: (c1="Cam")=true nên lấy $A1*B1=2*20$

dòng2: (c2="Cam")=false nên lấy 0 (theo cách của lấy của hàm IF)

dòng3: (c3="Cam")=true nên lấy $A3*B3=4*25$

Sau khi chạy hết các dòng, EXCEL sẽ dùng hàm SUM để tính tổng kết quả tính được ở từng dòng $=2*20+0+2*25=140$. Nếu trong công thức là hàm khác hàm SUM thì cách tính sẽ theo hàm đó.

Như vậy có 2 cách tính:

=SUMPRODUCT(A1:A3*B1:B3*(C1:C3="Cam")) và

=Sum(IF(C1:C3="Cam",A1:A3*B1:B3,0))

(*) Vậy tại sao không dùng là =SUM(A1:A3*B1:B3*(C1:C3="Cam"))

mà phải dùng hàm =SUMPRODUCT(A1:A3*B1:B3*(C1:C3="Cam")) ?

Chúng ta nhớ lại cấu trúc của hàm SUM là:

SUM(number1,number2, ...)

Còn SUMPRODUCT là

SUMPRODUCT(array1,array2,array3, ...)

number <> array

Nếu SUMPRODUCT(A1:A3*B1:B3*(C1:C3="Cam")) rồi ENTER là đúng vì đối số của nó phải là mảng - Array.

Nếu công thức =SUM(A1:A3*B1:B3*(C1:C3="Cam")) rồi ENTER kết quả là #VALUE! -lỗi vì A1:A3 là một array chứ không phải là một number.

Nếu nhấn tổ hợp **CTRL+SHIFT+ENTER**. Với cách làm này EXCEL sẽ phân tích như sau:

Xét trên từng dòng trong mảng (array)

dòng1: A1*B1*(c1="Cam")=2*20*True=2*20*1

dòng2: A2*B2*(c2="Cam")=3*10*False=3*10*0

dòng3: A3*B3*(c3="Cam")=2*25*True=4*25*1

Sau khi chạy hết các dòng, EXCEL sẽ dùng hàm SUM để tính tổng kết quả tính được ở từng dòng =2*20*1+3*10*0+4*25*1=140.

Vậy vẫn dùng được =SUM(A1:A3*B1:B3*(C1:C3="Cam")) với điều kiện nhấn tổ hợp phím **CTRL+SHIFT+ENTER**

Như vậy đến đây chúng ta có 3 cách tính:

=SUMPRODUCT(A1:A3*B1:B3*(C1:C3="Cam")) nhấn phím ENTER

=SUM(IF(C1:C3="Cam",A1:A3*B1:B3,0)) nhấn phím **CTRL+SHIFT+ENTER**

=SUM(A1:A3*B1:B3*(C1:C3="Cam")) nhấn phím **CTRL+SHIFT+ENTER**

Chúng có thể kết hợp rất nhiều điều kiện vào trong hàm thông qua phép toán logic nhân-và- And, cộng - hoặc - Or.

(*) Dùng hàm SUMPRODUCT hay dùng SUM kết hợp **CTRL+SHIFT+ENTER** đều cho ra được kết quả như nhau chính là do phép toán logic của bạn.

(*) Hàm SUMPRODUCT chỉ có thể tính tổng theo nhiều điều kiện

(*) Công thức mảng - Formula Array ngoài việc tính tổng có nhiều điều kiện còn làm rất nhiều phép tính khác do cách sử dụng hàm mà thôi.

BÀI TẬP

1. Tính doanh thu lớn nhất của mỗi khách hàng

	A	B	C	D	E
1					
2	TÊN KH	DOANH THU		TÊN KH	MAX DOANH THU
3	LAN	250		LAN	350
4	TUÂN	400		TUÂN	400
5	NHIÊN	600		NHIÊN	650
6	DUY	400		DUY	420
7	SƠN	150		SƠN	800
8	LINH	600		LINH	600
9	LAN	300			
10	TUÂN	350			
11	NHIÊN	650			
12	DUY	420			
13	SƠN	200			
14	LINH	550			
15	LAN	350			
16	TUÂN	300			
17	NHIÊN	580			
18	DUY	400			
19	SƠN	800			
20	LINH	200			

2. Lấy danh sách duy nhất.

3. Tìm giá trị lớn nhất thứ n trong một dãy số.

4. Mở tệp “TH NXT THEO ĐVQĐ.xls”. Lập sổ TH NXT và sổ TH NXT kết hợp đơn vị quy đổi.

5. Mở tệp “Ky thuật tạo bao cao.xls”. Thực hiện các bài tập theo yêu cầu tại sheet “Hệ thống đề bài”.